

## Poszukiwanie węzłów idealnych III – zastosowanie języka Maple V.4 do problemu ciasnego skręcenia lin<sup>1)</sup>

Sylwester PRZYBYŁ, Piotr PIERAŃSKI

Politechnika Poznańska, Wydział Fizyki Technicznej  
ul. Piotrowo 3, 60-965 Poznań  
e-mail: pieransk@phys.put.poznan.pl


Otrzymano 9 grudnia 1997 roku

**Streszczenie.** W artykule sformułowano problem optymalnego, ciasnego skręcenia dwu lin. Rozważono cztery różne struktury periodyczne (skrętki) określając parametry ich form optymalnych.

**Słowa kluczowe:** węzeł, węzeł idealny, skrętka

### 1. Wprowadzenie

Rozprostujmy dwie nieskończenie wiotkie, ale i nieskończenie twarde w swym kołowym przekroju liny, i umieśćmy je tak, by stykały się wzdłuż osi  $x$ , a następnie zaczniemy je skręcać z sobą tworząc periodyczną strukturę  $S$  o okresie  $p$ . Przyjmijmy, że średnice obu lin są identyczne i wynoszą  $D$ . Patrząc z boku, np. w kierunku osi  $y$ , widzimy jednowymiarową sieć równoodległych od siebie skrzyżowań – po dwa skrzyżowania na każdy okres skrętki  $S$ . Zadajmy pytanie: *jak wygląda konformacja skrętki, dla której iloraz  $L_C/D$  długości liny, zawartej w jednym okresie skrętki, i średnicy liny jest minimalny?*



Rys. 1. Najprostszy z czterech sposobów skręcenia ze sobą dwu lin: skrętka symetryczna  $S_1$

---

<sup>1)</sup> Praca wykonana w ramach grantu 8T11 F010 08p04.

Odpowiedź na postawione wyżej pytanie rzuca nieco światła na konformacje węzłów idealnych [1]. Znalezienie tej odpowiedzi nie jest jednak, jak się okazało, problemem trywialnym. W prezentowanej tu pracy rozwiązanie znajdujemy na drodze obliczeń numerycznych i symbolicznych wykonanych za pomocą pakietu Maple V.4. Wszystkie rysunki i wykresy ilustrujące przeprowadzone obliczenia zostały wykonane również za pomocą narzędzi tego pakietu.

### 1.1. Założenia modelu

Założmy, zgodnie z tym co obserwujemy w doświadczeniu, że osie skręcanych lin, A i B, są liniami śrubowymi o tym samym skoku  $p$ , przesuniętymi w fazie o  $\pi$ . Przyjmijmy, co uprości prezentowane wyrażenia (nie ograniczając ich ogólności), że średnica obu lin  $D = 1$ .

Równania opisujące kolejno pierwszą i drugą linię przedstawiają się więc w postaci parametrycznej następująco:

$$\begin{aligned} X_A &= pt, \\ Y_A &= R_A \sin(2\pi t), \\ Z_A &= R_A \cos(2\pi t), \end{aligned} \quad (1.1)$$

$$\begin{aligned} X_B &= pt, \\ Y_B &= -R_B \sin(2\pi t), \\ Z_B &= -R_B \cos(2\pi t), \end{aligned} \quad (1.2)$$

gdzie  $R_A$  i  $R_B$  oznaczają promienie linii śrubowych A i B, a  $p$  oznacza skok tych linii.

Całkowita długość  $L_C$  obu lin w jednym okresie skrętki wynosi:

$$L_C = \sqrt{(2\pi R_A)^2 + p^2} + \sqrt{(2\pi R_B)^2 + p^2}. \quad (1.3)$$

Poszukujemy takich wartości parametrów obu linii śrubowych, a więc ich promieni  $R_A$ ,  $R_B$  i skoku  $p$ , dla których  $L_C$  osiąga minimum, przy zachowaniu warunku, że liny nie przekrywają się ani między sobą, ani same ze sobą. Pojawienie się takich przekryć monitorują odpowiednio zdefiniowane funkcje  $d_{AB}(\tau)$  i  $d_{AA}(\tau)$ .

### 1.2. Definicja funkcji $d_{AB}(\tau)$


Wybermy na linii śrubowej B pewien punkt  $P_B = [X_B(t_p), Y_B(t_p), Z_B(t_p)]$ , wyznaczony przez równanie (1.2) dla wartości parametru  $t = t_p$ . Zdefiniujmy funkcję  $d_{AB}(\tau)$  określającą odległość między wybranym punktem i punktami linii śrubowej A wyznaczonymi przez równanie (1.1) dla wartości parametru  $t = t_p + \tau$ :

$$\begin{aligned} d_{AB}(\tau) &= \\ &= \sqrt{[X_B(t_p) - X_A(t_p + \tau)]^2 + [Y_B(t_p) - Y_A(t_p + \tau)]^2 + [Z_B(t_p) - Z_A(t_p + \tau)]^2}. \end{aligned} \quad (1.4)$$

Ponieważ wartość  $t_p$ , określającą punkt  $P_B$ , możemy wybrać dowolnie, przyjmijmy, że wynosi ona 0. Wtedy

$$d_{AB}(\tau) = \sqrt{[X_B(0) - X_A(\tau)]^2 + [Y_B(0) - Y_A(\tau)]^2 + [Z_B(0) - Z_A(\tau)]^2}. \quad (1.5)$$

Dla  $\tau = 0$  funkcja  $d_{AB}$  ma, w zależności od rozważanego modelu skrętki, wartość równą lub większą od 1. Jej zachowanie w otoczeniu  $\tau = 0$  pozwala określić, czy w tworzonej skrętce nie pojawiają się przekrycia między skręcanymi linami: jak długo w otoczeniu  $\tau = 0$  nie ma punktów, dla których wartość funkcji  $d_{AB}$  spadałaby poniżej 1, tak długo skręcane z sobą linie nie przekrywają się.



Rys. 2. Funkcja  $d_{AB}(\tau)$  określa odległość między wybranym punktem jednej z osi śrubowych a punktami drugiej osi

### 1.3. Definicja funkcji $d_{AA}(\tau)$

Wybermy na linii śrubowej A pewien punkt  $P_A = [X_A(t_p), Y_A(t_p), Z_A(t_p)]$ , wyznaczony przez równanie (1.1) dla wartości parametru  $t = t_p$ . Zdefiniujmy funkcję  $d_{AA}(\tau)$ , określającą odległość między wybranym punktem i punktami linii śrubowej A, wyznaczonymi przez równanie (1.1) dla wartości parametru  $t = t_p + \tau$ :

$$d_{AA}(\tau) = \sqrt{[X_A(t_p) - X_A(t_p + \tau)]^2 + [Y_A(t_p) - Y_A(t_p + \tau)]^2 + [Z_A(t_p) - Z_A(t_p + \tau)]^2}. \quad (1.6)$$

Ponieważ wartość  $t_p$  określającą punkt  $P_B$  możemy wybrać dowolnie, przyjmijmy, że wynosi ona 0. Wtedy

$$d_{AA}(\tau) = \sqrt{[X_A(0) - X_A(\tau)]^2 + [Y_A(0) - Y_A(\tau)]^2 + [Z_A(0) - Z_A(\tau)]^2}. \quad (1.7)$$

Zachowanie się funkcji  $d_{AA}$  w pobliżu  $\tau = 1$  pozwala określić, czy w tworzonej skrajce nie pojawiają się przekrycia liny A z nią samą; przekryć takich nie ma, jeśli w otoczeniu  $\tau = 1$  nie ma punktów, dla których wartość funkcji  $d_{AB}$  spadałaby poniżej 1.

Podobny test trzeba przeprowadzić dla liny B.

## 2. $S_1$ – skrajka symetryczna o jednej, prostej linii styku

Istnieją przynajmniej cztery różne sposoby skrajania z sobą dwu lin tak, by utworzona skrajka była liniową strukturą periodyczną. Najprostszym z rozważanych niżej sposobów skrajania jest ten, w którym jedynym zbiorem punktów styczności obu lin pozostaje linia prosta, wzdłuż której liny te stykały się w swej konformacji wyjściowej, gdy były wyprostowane. Oznaczmy ten typ skrajki przez  $S_1$ . Łatwo zauważyć, że w skrajce tego typu muszą zachodzić równości

$$R_A = R_B = R = \frac{1}{2}. \quad (2.1)$$

Ponieważ

$$\begin{aligned} X_B(0) &= 0, \\ Y_B(0) &= 0, \\ Z_B(0) &= -R, \end{aligned} \quad (2.2)$$

więc


$$\begin{aligned} d_{AB}(\tau) &= \sqrt{X_A^2(\tau) + Y_A(\tau)^2 + [-R - Z_A(\tau)]^2}, \\ d_{AB}(\tau) &= \sqrt{(p\tau)^2 + [R \sin(2\pi\tau)]^2 + [R \cos(2\pi\tau) + R]^2}, \end{aligned} \quad (2.3)$$

czyli, ze względu na (2.1), mamy

$$d_{AB}(\tau) = \sqrt{\frac{1}{2} \cos(2\pi\tau) + \frac{1}{2} + p^2\tau^2}. \quad (2.4)$$

Kształt funkcji  $d_{AB}(\tau)$  zależy od wartości parametru  $p$  określającego okres skrajki (zob. rys. 3). Jak widać, istnieje pewna minimalna wartość  $p$ , powyżej której w otoczeniu  $\tau = 0$  funkcja  $d_{AB}(\tau) > 1$ . Poniżej tej wartości funkcja  $d_{AB}(\tau)$  staje się w otoczeniu  $\tau = 0$  mniejsza od 1, co oznacza, że w skrajce pojawiają się przekrycia lin. Ponadto  $d_{AB}(\tau) \geq 1$  tak długo, jak

$$\left\{ \frac{d^2}{d\tau^2} d_{AB}(\tau) \right\}_{\tau=0} \geq 0. \quad (2.5)$$

Rys. 3. Postać funkcji  $d_{AB}(\tau)$  w otoczeniu  $\tau = 0$  w zależności od skoku  $p$  skrętki

Wykorzystując ten warunek, możemy łatwo wyznaczyć minimalny skok skrętki:


$$p = \pi. \quad (2.6)$$

Długość  $L_C$  liny zawartej w jednym okresie skrętki wynosi tu więc

$$L_C = 2\sqrt{(2\pi R)^2 + p^2} = 2\sqrt{2}\pi, \quad (2.7)$$

skąd, ponieważ  $D = 1$ , poszukiwane minimum ilorazu długości liny zawartej w jednym okresie skrętki do jej średnicy wynosi tu

$$\frac{L_C}{D} = 8,885766\dots \quad (2.8)$$



Rys. 4. Proces skręcania dwu lin w skrętce typu  $S_1$ . Proces ustaje, gdy dalsze skręcanie powodowałoby pojawienie się przekryć

Przeprowadzone rozważania zostały oparte na wynikach obliczeń symbolicznych wykonanych za pomocą pakietu Maple V.4. Opisane wyznaczenie minimalnego skoku i odpowiadającej mu minimalnej długości linii w jednym okresie skrętki realizowane jest przez następujący, krótki program:

```

> restart;
> xA:=(t)->p*t;
> yA:=(t)->R*sin(2*Pi*t);
> zA:=(t)->R*cos(2*Pi*t);
>
> xB:=(t)->p*t;
> yB:=(t)->-R*sin(2*Pi*t);
> zB:=(t)->-R*cos(2*Pi*t);
>
> dAB:=
> (tau)->sqrt((xB(tau)-xA(0))^2+(yB(tau)-yA(0))^2+(zB(tau)-
> zA(0))^2);
>
> dABprim:=unapply(simplify(diff(dAB(tau),tau),tau));
>
> dABbis:=unapply(simplify(diff(dABprim(tau),tau),tau));
>
> eq3:=dABbis(tau)=0;
>
> R:=1/2;
> tau:=0;
>
> eq3:=simplify(eq3);
> sol:=solve(eq3);
>
> p:=max(sol[1],sol[2]);
> Lc:=sqrt((2*Pi*R)^2+p^2)+sqrt((2*Pi*R)^2+p^2);

```

### 3. $S_2$ – skrętka symetryczna o dwu śrubowych liniach styku

Założmy, jak to uczyniliśmy poprzednio, że osie skręcanych lin są przesuniętymi w fazie o  $\pi$  liniami śrubowymi o identycznych promieniach  $R_A = R_B = R$ , jednak zrezygnujemy z warunku, że  $R = 0,5$ ; przyjmujemy więc, że  $R$  może być większe od 0,5. Konstruowana przy tych założeniach skrętka, oznaczmy ją przez  $S_2$ , ma strukturę inną niż  $S_1$ : tworzące ją liny oplatają się tu nie wokół prostej, lecz wokół wirtualnego cylindra o promieniu  $R - 1$  (zob. rys. 4). Skręcane liny nie mają początkowo żadnych punktów styku z sobą (stykają się wyłącznie z powierzchnią wirtualnego cylindra, na którym są oplatanie). Gdy jednak malejący skok skrętki osiąga pewną progową wartość, w skrętce pojawiają się dwie symetryczne, śrubowe linie punktów styku; dalsze skracanie skoku staje się niemożliwe. Utworzona skrętka  $S_2$  jest podobna do  $S_1$ , lecz w odróżnieniu od niej ma pusty środek. Czy w tej sytuacji iloraz  $L_C/D$  może okazać się mniejszy niż dla skrętki  $S_1$ ? Okazuje się, że tak. Dowód przedstawiamy poniżej.

Zdefiniujmy, jak poprzednio (zob. równanie (1.4)), funkcję  $d_{AB}(\tau)$  określającą odległość między wybranym punktem  $P_B = [X_B(t_p), Y_B(t_p), Z_B(t_p)]$  linii śrubowej B i punktami osi śrubowej A położonymi w otoczeniu punktu  $P_A = [X_A(t_p), Y_A(t_p), Z_A(t_p)]$ . Funkcja  $d_{AB}(\tau)$  przy założonym tu  $R > 0,5$ , przyjmuje dla  $\tau = 0$  wartości większe od 1, jednak, przy malejącym  $p$  w funkcji tej pojawiają się w pobliżu  $\tau = 0$  dwa symetrycznie rozmieszczone minima (zob. rys. 5). W chwili, gdy wartość  $d_{AB}(\tau)$  osiągnie w tych minimach wartość 1, skręcane liny zetkną się. Poszukiwanie minimalnej wartości skoku skrętki  $S_2$ , sprowadza się do wyznaczenia położenia ekstremów funkcji  $d_{AB}(\tau)$ , a więc do rozwiązania równania

$$\frac{d}{d\tau} d_{AB}(\tau) = 0, \quad (3.1)$$

z którego, przy dostatecznie małym  $p$ , prócz oczywistego pierwiastka  $\tau_0 = 0$  powinniśmy otrzymać dwa, leżące w pobliżu, symetrycznie rozłożone pierwiastki  $\tau_1 = -\tau_2$ . Pierwiastki te opisują położenia dwu minimów funkcji  $d_{AB}(\tau)$  pojawiających się przy malejącym  $p$ . Znając  $\tau_1$  i  $\tau_2$  można wyznaczyć tę wartość  $p$ , dla której

$$d_{AB}(\tau_{1,2}) = 1, \quad (3.2)$$


a następnie obliczyć długość liny w jednym okresie skrętki:

$$L_C = 2\sqrt{(2\pi R)^2 + p^2}. \quad (3.3)$$

Zauważmy, że w podanym wyżej wzorze  $p$  jest również zależne od  $R$ . Ze względu na nieliniowość równań (3.1) i (3.2), ich analityczne rozwiązanie nie jest możliwe. W efekcie zależności  $L_C(R)$  nie daje się podać w jawnej postaci.

Wspomnianą zależność można jednak bez trudu wykreślić korzystając z procedury wykreślenia przebiegu funkcji danej w postaci parametrycznej. Podany niżej program pokazuje jedną z możliwości realizacji tego ciekawego zadania.

```
>restart;
> with (plots) :
```




**Rys. 5.** Konstrukcja skrętki typu  $S$ . Dwie liny okrężane są symetrycznie wokół wirtualnego cylindra. Skok tworzonej skrętki maleje do momentu, gdy liny zetkną się (dla przejrzystości rysunek wykonano przy średnicy wirtualnego cylindra większej od optymalnej)

```

>
> xA:=(t)->p*t;
> yA:=(t)->R*sin(2*Pi*t);
> zA:=(t)->R*cos(2*Pi*t);
>
> xB:=(t)->p*t;
> yB:=(t)->-R*sin(2*Pi*t);
> zB:=(t)->-R*cos(2*Pi*t);
>
> dAB:=(tau)->sqrt((xB(tau)-xA(0))^2+(yB(tau)-yA(0))^2+
> (zB(tau)-zA(0))^2);
> dABprim:=unapply(diff(dAB(tau),tau));
>
> eq1:=dAB(tau)=1;
> eq2:=dABprim(tau)=0;

```






Rys. 6. Postać funkcji  $d_{AB}(\tau, p)$  przy  $R = 0,6$ . Wykres obcięty od dołu na wartości  $d_{AB} = 1$  wyznaczającej pojawienie się styku między liniami

```
>
> p:=solve(eq1,p)[1];
> R:=solve(eq2,R)[1];
>
> Lc:=2*sqrt((2*Pi*R)^2+p^2);
>
> f:=[R,Lc,tau=0.01..0.33]:
>
> plot(f,color=black,axesfont=[TIMES,ROMAN,18]);
```

Wynik jest przedstawiony na rys. 7. Jak widać, istnieje taka wartość  $R$ , dla której długość linii w jednym okresie skrętki typu  $S_2$  zostaje zminimalizowana. Wartość tę można wyznaczyć numerycznie, korzystając z procedury *fsolve*, na przykład w następujący sposób:

```
>restart;
> xA:=(t)->p*t;
> yA:=(t)->R*sin(2*Pi*t);
> zA:=(t)->R*cos(2*Pi*t);
>
> xB:=(t)->p*t;
> yB:=(t)->-R*sin(2*Pi*t);
> zB:=(t)->-R*cos(2*Pi*t);
>
> dAB:=(tau)->sqrt((xB(tau)-xA(0))^2+(yB(tau)-yA(0))^2
> +(zB(tau)-zA(0))^2);
> dABprim:=unapply(simplify(diff(dAB(tau),tau)));
>
> eq1:=dAB(tau)=1;
> eq2:=dABprim(tau)=0;
>
> p:=solve(eq1,p)[1];
> R:=solve(eq2,R)[1];
>
> Lc:=2*sqrt((2*Pi*R)^2+p^2);
> Lcprim:=diff(Lc,tau);
```



Rys. 7. Zależność  $L_c$  od promienia  $R$  dla skrętki typu  $S_2$

```
> eq3:=Lcprim=0;
>
> tau:=fsolve(eq3,tau=0.1..0.33);
> Rmin:=evalf(R);
> pmin:=evalf(p);
> Lcmin:=evalf(Lc);
```

Wykonanie ostatnich trzech instrukcji tego prostego programu przynosi oczekiwany wynik:

$$R = 0,522935, \quad p = 2,69670, \quad L_c = 8,50131. \quad (3.4)$$

Jak widać, skręcanie lin w skrętke typu  $S_2$  zdecydowanie zmniejsza zużycie liny.

#### 4. $S_3$ – skrętka asymetryczna o jednej śrubowej linii styku

Rozważmy kolejną możliwość. Załóżmy jak poprzednio, że osie skręcanych lin są przesuniętymi w fazie o  $\pi$  liniami śrubowymi, jednak o nierównych promieniach  $R_A \neq R_B$ . Przyjmijmy dalej, podobnie jak to uczyniliśmy dla przypadku  $S_1$ , że suma promieni

$$R_A + R_B = 1, \quad (4.1)$$

tzn., że skręcane liny pozostają ze sobą w stałym kontakcie, stykając się wzdłuż jednej, śrubowej linii styku. Dla danej wartości, np. wartości  $R_A$ , istnieje tu minimalny skok skrętki, poniżej którego liny skręcane ze sobą zaczynają się przekrywać. Przekrycia pojawiają się, gdy, podobnie jak to się działo w przypadku skrętki typu  $S_1$ , w otoczeniu  $\tau = 0$  (gdzie  $d_{AB} = 1$ ), w funkcji  $d_{AB}(\tau)$  rozwiną się dwa, symetrycznie rozmieszczone minima.

Chwilę pojawienia się minimów zwiastuje zerowanie się w punkcie  $\tau = 0$  pochodnej drugiego rzędu funkcji  $d_{AB}(\tau)$ .

Poniższy program obliczeń symbolicznych realizuje zadanie poszukiwania minimalnego skoku skrętki przy zadanej wartości  $R_A$ .

```
>restart;
> xA:=(t)->p*t;
> yA:=(t)->RA*sin(2*Pi*t);
> zA:=(t)->RA*cos(2*Pi*t);
>
> xB:=(t)->p*t;
> yB:=(t)->-RB*sin(2*Pi*t);
> zB:=(t)->-RB*cos(2*Pi*t);
>
> dAB:=sqrt((xB(tau)-xA(0))^2+(yB(tau)-yA(0))^2+(zB(tau)-
> zA(0))^2);
> dABprim:=diff(dAB,tau);
> dABbis:=diff(dABprim,tau);
>
> dABbis0:=subs(tau=0,dABbis);
>
> RB:=1-RA;
>
> eq1:=dABbis0=0;
>
> solP:=solve(eq1,p);
> p:=solP[1];
```

Wynikiem wykonania tego programu jest prosta formuła:

$$p = 2\pi\sqrt{R_A(1-R_A)}, \quad (4.2)$$

która określa minimalny skok skrętki przy zadanej wartości  $R_A$ .


Na rys. 8 przedstawiono tworzone w ten sposób optymalne skrętki. Jak widać, skracanie skoku skrętki  $S_3$ , okupione jest tu powiększeniem asymetrii jej struktury.

Skracanie skoku skrętki jest korzystne z punktu widzenia zużycia liny. Wykreślenie zależności  $L_C(p)$  jasno na to wskazuje (zob. rys.9). Rys. 9 został wykonany przez wykonanie dwu dodatkowych instrukcji:

```
>Lc:=sqrt((2*Pi*RA)^2+p^2)+sqrt((2*Pi*RB)^2+p^2);
> plot(Lc, RA=0.5..1, color=black, axesfont=[TIMES,ROMAN,15]);
```

Zależność  $L_C(R_A)$  wykreślono tu dla  $R_A$ , zmieniającego się od 0,5 do 1, jednak, jak to jasno widać z rys. 8, wartość  $R_A = 1$  nie może być w oczywisty sposób osiągnięta, gdyż wcześniej pojawiają się w tworzonej skrętce przekrycia liny A z nią samą. Chwilę zahamowania procesu skracania skoku skrętki typu  $S_3$  można określić badając funkcję  $d_{AA}(\tau)$ . Rys. 10 przedstawia jej kształt przy  $R_A$  zmieniającym się od 0,5 do 1. Wykres funkcji  $d_{AA}(\tau)$  otrzymujemy kontynuując rozpoczęte wyżej obliczenia:

```
>dAA:=(xA(tau)-xA(0))^2+(yA(tau)-yA(0))^2+(zA(tau)-zA(0))^2;
>
> plot3d(dAA,tau=-2..2,RA=0.5..1,
> shading=NONE,
> style=PATCH,
> lightmodel=light4,
> axes=BOXED,
```




Rys. 8. Skrętka typu  $S_3$  dla różnych wartości promienia  $R_A$

```
> axesfont=[TIMES,ROMAN,15],
> orientation=[111,25],
> grid=[100,100],
> view=[-2..2,0.5..1,0..10]);
>
```


Jak widać, w funkcji  $d_{AA}(\tau)$  pojawiają się przy rosnącym  $R_A$  dwa lokalne minima zlokalizowane w pobliżu  $\tau = \pm 1$ . W chwili, gdy minima te osiągną krytyczną wartość 1 (wartości mniejsze od 1 zwiastują pojawienie się przekryć), dalsze skracanie skoku skrętki staje się niemożliwe.

Wyznaczenia punktu osiągnięcia minimalnej wartości skoku i określenia innych parametrów optymalnej skrętki typu  $S_3$  realizuje następujący program:

```
>restart;
> xA:=(t)->p*t;
```



Rys. 9. Zależność długości  $L_C$  liny zawartej w jednym okresie optymalnej skrętki typu  $S_3$  od promienia  $R_A$



Rys. 10. Zachowanie się funkcji  $d_{AA}(\tau)$  przy zmieniającej się wartości promienia  $R_A$

```
> yA:=(t)->RA*sin(2*Pi*t);
> zA:=(t)->RA*cos(2*Pi*t);
>
> xB:=(t)->p*t;
> yB:=(t)->-RB*sin(2*Pi*t);
> zB:=(t)->-RB*cos(2*Pi*t);
>
```

```

> dAB:=sqrt((xB(tau)-xA(0))^2+(yB(tau)-yA(0))^2+(zB(tau)-
> zA(0))^2);
> dABprim:=simplify(diff(dAB,tau));
> dABbis:=simplify(diff(dABprim,tau));
> dABbis0:=subs(tau=0,dABbis);
> dABbis0:=simplify(dABbis0);
>
> RB:=1-RA;
>
> eq1:=dABbis0=0;
>
> solP:=solve(eq1,p);
> p:=solP[1];
>
> dAA:=sqrt((xA(tau)-xA(0))^2+(yA(tau)-yA(0))^2+(zA(tau)-
> zA(0))^2);
> dAAprime:=diff(dAA,tau);
> dAAbis:=diff(dAAprime,tau);
>
> eq2:=dAA=1;
> solA:=solve(eq2,RA);
> RA:=solA[1];
>
> eq3:=dAAprime=0;
>
> tau:=fsolve(eq3,tau=0.8..1.2);
>
> RAmin=evalf(RA);
> RBmin=evalf(RB);
> pmin=evalf(p);
> Lc:=sqrt((2*Pi*RA)^2+p^2)+sqrt((2*Pi*RB)^2+p^2);
> Lcmin=evalf(Lc);
>

```

W wyniku jego wykonania otrzymujemy następujące wartości parametrów optymalnej skrętki typu  $S_3$ :

$$R_A = 0,9732568, \quad R_B = 0,0267432, \quad p = 1,01368, \quad L_C = 7,22611. \quad (4.3)$$

## 5. $S_4$ – skrętka asymetryczna o dwu śrubowych liniach styku

Podobnie, jak to uczyniliśmy, przechodząc od rozwiązania typu  $S_1$  do rozwiązania typu  $S_2$ , możemy rozważyć uogólnioną skrętkę typu  $S_3$ , rezygnując z zachowania warunku  $R_A + R_B = 1$ . Przeprowadzone rozważania wykazują, że ten typ skrętki, oznaczmy go symbolem  $S_4$ , cechuje się jeszcze nieco lepszymi parametrami:

$$p = 1,013859, \quad L_C = 7,201261. \quad (5.1)$$

osiąganymi w sytuacji, gdy promienie osi śrubowych skręcanych z sobą lin wynoszą odpowiednio:

$$R_A = 0,967031, \quad R_B = 0,037743, \quad (5.2)$$

a więc, gdy ich suma jest nieco większa od jedności.

Przypadek ten, nieco bardziej złożony, rozważymy szczegółowo w kolejnej publikacji naszego cyklu poświęconego poszukiwaniu węzłów złożonych.

### **Podziękowania**

Dziękujemy Panu Andrzejowi Stasiakowi za liczne dyskusje.

### **Literatura**

- [1] Katritch, V. (et al), *Nature* 384 (1996), 142.
- [2] Pierański, P., Poszukiwanie węzłów idealnych, *Pro Dialog* 5 (1996), 111–120.
- [3] Adams, C. C., *The Knot Book*, W. H. Freeman and Co., New York 1994.